
D04 Procedural vs Denotative Style 
 
 
 
      ⍝ Procedural vs denotative styles 
       
      ⍝ Different mindsets - both have their place 
       
      ⍝ Procedural: aware of sequence 
      ⍝ Denotative: atemporal definitions 
       
      ⍝ Example: find the shortest path through a graph (network) 
       
      ⍝   ┌───1───┐ 
      ⍝   ↓   ↓   ↑ 
      ⍝   2←─→3──→4 
      ⍝       ↑   ↓ 
      ⍝       └───5 
       
      g←(2 3)(3)(2 4)(1 5)(3)     ⍝ directed graph 
       
      g pathP 2 1                 ⍝ single vertex to vertex path 
2 3 4 1 
       
      ed ggg                      ⍝ parallel breadth-first search 
      ed ggg                      ⍝ parallel breadth-first search 
       
      ⍪ggg 
┌───────────────────────────┐ 
│[0]-----[⎕]-----[⎕]-----[⎕]│ 
│ |       |       |       | │ 
│ |       |       |       | │ 
│ |       |       |       | │ 
│[⎕]-----[⎕]     [⎕]-----[⎕]│ 
│ |                       | │ 
│ |                       | │ 



│ |                       | │ 
│[⎕]-----[⎕]     [⎕]-----[⎕]│ 
│ |       |       |       | │ 
│ |       |       |       | │ 
│ |       |       |       | │ 
│[⎕]-----[⎕]-----[⎕]-----[⎕]│ 
├───────────────────────────┤ 
│[0]─────[1]-----[⎕]-----[⎕]│ 
│ │       |       |       | │ 
│ │       |       |       | │ 
│ │       |       |       | │ 
│[1]-----[⎕]     [⎕]-----[⎕]│ 
│ |                       | │ 
│ |                       | │ 
│ |                       | │ 
│[⎕]-----[⎕]     [⎕]-----[⎕]│ 
│ |       |       |       | │ 
│ |       |       |       | │ 
│ |       |       |       | │ 
│[⎕]-----[⎕]-----[⎕]-----[⎕]│ 
├───────────────────────────┤ 
│[0]─────[1]─────[2]-----[⎕]│ 
│ │       │       |       | │ 
│ │       │       |       | │ 
│ │       │       |       | │ 
│[1]─────[2]     [⎕]-----[⎕]│ 
│ │                       | │ 
│ │                       | │ 
│ |                       | │ 
│[2]-----[⎕]     [⎕]-----[⎕]│ 
│ |       |       |       | │ 
│ |       |       |       | │ 
│ |       |       |       | │ 
│[⎕]-----[⎕]-----[⎕]-----[⎕]│ 
├───────────────────────────┤ 
│[0]─────[1]─────[2]─────[3]│ 



│ │       │       │       | │ 
│ │       │       │       | │ 
│ │       │       │       | │ 
│[1]─────[2]     [3]-----[⎕]│ 
│ │                       | │ 
│ │                       | │ 
│ |                       | │ 
│[2]─────[3]     [⎕]-----[⎕]│ 
│ │       |       |       | │ 
│ │       |       |       | │ 
│ │       |       |       | │ 
│[3]-----[⎕]-----[⎕]-----[⎕]│ 
├───────────────────────────┤ 
│[0]─────[1]─────[2]─────[3]│ 
│ │       │       │       │ │ 
│ │       │       │       │ │ 
│ │       │       │       │ │ 
│[1]─────[2]     [3]─────[4]│ 
│ │                       | │ 
│ │                       | │ 
│ |                       | │ 
│[2]─────[3]     [⎕]-----[⎕]│ 
│ │       │       |       | │ 
│ │       │       |       | │ 
│ │       │       |       | │ 
│[3]─────[4]-----[⎕]-----[⎕]│ 
├───────────────────────────┤ 
│[0]─────[1]─────[2]─────[3]│ 
│ │       │       │       │ │ 
│ │       │       │       │ │ 
│ │       │       │       │ │ 
│[1]─────[2]     [3]─────[4]│ 
│ │                       │ │ 
│ │                       │ │ 
│ |                       │ │ 
│[2]─────[3]     [⎕]-----[5]│ 



│ │       │       |       | │ 
│ │       │       |       | │ 
│ │       │       |       | │ 
│[3]─────[4]─────[5]-----[⎕]│ 
├───────────────────────────┤ 
│[0]─────[1]─────[2]─────[3]│ 
│ │       │       │       │ │ 
│ │       │       │       │ │ 
│ │       │       │       │ │ 
│[1]─────[2]     [3]─────[4]│ 
│ │                       │ │ 
│ │                       │ │ 
│ |                       │ │ 
│[2]─────[3]     [6]─────[5]│ 
│ │       │       │       │ │ 
│ │       │       │       │ │ 
│ │       │       │       │ │ 
│[3]─────[4]─────[5]─────[6]│ 
└───────────────────────────┘ 
             
  



      pathP                 ⍝   P r o c e d u r a l   c o d i n g 
{                           ⍝ Shortest path from/to ⍵ in graph ⍺ 
    fm to←⍵                 ⍝ entry/exit vertices                
    tree←¯2+(⍳⍴⍺)∊fm        ⍝ initial spanning tree              
    free←⍺{⍺⍺[⍵]∩¨⊂⍸⍺=¯2}   ⍝ free vertices in ⍵                 
    {                       ⍝ find partial spanning tree:        
        ⍵≡⍬:⍬               ⍝ no vertices left: stitched         
        ∨/to∊⍵:⍬{           ⍝ found target: path from tree:      
            ⍵<0:⍺           ⍝ root or unvisited vertex: done     
            (⍵,⍺)∇ ⍵⊃tree   ⍝ accumulated path to next vertex    
        }1↑⍵∩to             ⍝ found target: finished             
        next←tree free ⍵    ⍝ next vertices to visit             
        back←⊃,/⍵+0×next    ⍝ back links                         
        wave←⊃,/next        ⍝ vertex wave front                  
        tree[wave]←back     ⍝ SET back links in tree !!!         
        ∇∪wave              ⍝ advance wave front                 
    }fm                     ⍝ from starting vertex               
}                                                                     
       

      pathD                 ⍝   D e n o t a t i v e   c o d i n g 
{                           ⍝ Shortest path fm/to ⍵ in graph ⍺ 
    fm to←⍵                 ⍝ entry-exit vertex vectors        
    tree←¯2+(⍳⍴⍺)∊fm        ⍝ null spanning tree               
    free←⍺{⍺⍺[⍵]∩¨⊂⍸⍺=¯2}   ⍝ free vertices in ⍵               
    tree{                   ⍝ find partial spanning tree:      
        ⍵≡⍬:⍬               ⍝ no vertices left: no path        
        ∨/to∊⍵:⍬(⊃∘⍺){      ⍝ found target: path from tree:    
            ⍵<0:⍺           ⍝ root or unvisited vertex: done   
            (⍵,⍺)∇ ⍺⍺ ⍵     ⍝ accumulated path to next vertex  
        }1↑⍵∩to             ⍝ found vertex ⍺                   
        next←⍺ free ⍵       ⍝ next vertices to visit           
        back←⊃,/⍵+0×next    ⍝ back links                       
        wave←⊃,/next        ⍝ vertex wave front                
        tree←back@wave⊢⍺    ⍝ tree with back links             
        tree ∇∪wave         ⍝ advanced wave front              
    }fm                     ⍝ from starting vertex             
}                                                              



       
      dtb∘rmcm∘⎕cr¨ 'pathP' 'pathD' 
┌─────────────────────────┬─────────────────────────┐ 
│pathP←{                  │pathD←{                  │ 
│    fm to←⍵              │    fm to←⍵              │ 
│    tree←¯2+(⍳⍴⍺)∊fm     │    tree←¯2+(⍳⍴⍺)∊fm     │ 
│    free←⍺{⍺⍺[⍵]∩¨⊂⍸⍺=¯2}│    free←⍺{⍺⍺[⍵]∩¨⊂⍸⍺=¯2}│ 
│    {                    │    tree{                │ 
│        ⍵≡⍬:⍬            │        ⍵≡⍬:⍬            │ 
│        ∨/to∊⍵:⍬{        │        ∨/to∊⍵:⍬(⊃∘⍺){   │ 
│            ⍵<0:⍺        │            ⍵<0:⍺        │ 
│            (⍵,⍺)∇ ⍵⊃tree│            (⍵,⍺)∇ ⍺⍺ ⍵  │ 
│        }1↑⍵∩to          │        }1↑⍵∩to          │ 
│        next←tree free ⍵ │        next←⍺ free ⍵    │ 
│        back←⊃,/⍵+0×next │        back←⊃,/⍵+0×next │ 
│        wave←⊃,/next     │        wave←⊃,/next     │ 
│        tree[wave]←back  │        tree←back@wave⊢⍺ │ 
│        ∇∪wave           │        tree ∇∪wave      │ 
│    }fm                  │    }fm                  │ 
│}                        │}                        │ 
└─────────────────────────┴─────────────────────────┘ 
       
 
 


